# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Patrick Ulbrich <zulu99 at gmx . net>

pkgname=mailnag-goa-plugin
pkgver=2.0.0
pkgrel=4
pkgdesc='Mailnag GNOME Online Accounts plugin'
arch=('any')
url='https://github.com/pulb/mailnag-goa-plugin'
license=('GPL')
depends=('gnome-online-accounts' 'mailnag')
source=("https://github.com/pulb/$pkgname/archive/v$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('8adfbd21d9c42243381856c2d46b1c97ec35c48fad5a0057e7b7e5c54bf7a45c')

build() {
  cd $pkgname-$pkgver
  python setup.py build
}

package() {
  cd $pkgname-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1 --skip-build
}
